G05EDF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G05EDF sets up the reference vector R for a binomial distribution of the number of successes in n trials, each with probability of success p.

2 Specification

```
SUBROUTINE G05EDF(N, P, R, NR, IFAIL)
INTEGER           N, NR, IFAIL
real              P, R(NR)
```

3 Description

G05EDF sets up a reference vector for use in G05EYF. Together these routines produce random numbers from the distribution defined by:

$$\begin{aligned} P(I=i) &= \frac{n!}{i!(n-i)!p^i(1-p)^{n-i}} && \text{if } i=0,\ldots,n, \\ P(I=i) &= 0 && \text{otherwise.} \end{aligned}$$

The reference array is found by a recurrence relation if $np(1-p) < 50$; otherwise Stirling's approximation is used.

4 References

[1] Knuth D E (1981) *The Art of Computer Programming (Volume 2)* Addison–Wesley (2nd Edition)

[2] Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* Griffin (3rd Edition)

5 Parameters

1: N — INTEGER *Input*

On entry: the number of trials, n, of the distribution.

Constraint: $\mathrm{N} \geq 0$.

2: P — ***real*** *Input*

On entry: the probability of success, p, of the distribution.

Constraint: $0 \leq \mathrm{P} \leq 1$.

3: R(NR) — ***real*** array *Output*

On exit: the reference vector.

4: NR — INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05EDF is called.

Suggested value: $\mathrm{NR} = 20 + 20\sqrt{\mathrm{N} \times \mathrm{P}(1-\mathrm{P})}$ approximately (for optimum efficiency in G05EYF).

Constraint:

$$\begin{aligned} \mathrm{NR} &> \min(\mathrm{N}, \mathrm{INT}[\mathrm{N} \times \mathrm{P} + 7.15\sqrt{\mathrm{N} \times \mathrm{P}(1-\mathrm{P})} + 1]) \\ &- \max(0, \mathrm{INT}[\mathrm{N} \times \mathrm{P} - 7.15\sqrt{\mathrm{N} \times \mathrm{P}(1-\mathrm{P})} - 7.15]) + 4. \end{aligned}$$

5: IFAIL — INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, N < 0.

IFAIL = 2

On entry, P < 0,
or P > 1.

IFAIL = 3

On entry, NR is too small (see Section 5).

7 Accuracy

Not applicable.

8 Further Comments

The time taken by the routine increases with NR.

9 Example

The example program sets up a reference vector for a binomial distribution with $n = 100$ and $p = 0.5$; it then prints the first five pseudo-random numbers generated by G05EYF, after initialisation by G05CBF.

The generator mechanism used is selected by an initial call to G05ZAF.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G05EDF Example Program Text
*     NAG Fortran SMP Library, Release 2.  NAG Copyright 2000.
*     .. Parameters ..
      INTEGER          N
      DOUBLE PRECISION P
      INTEGER          NR
      PARAMETER        (N=100,P=0.5D0,NR=125)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      INTEGER          I, IFAIL, IX
*     .. Local Arrays ..
      DOUBLE PRECISION R(NR)
*     .. External Functions ..
      INTEGER          G05EYF
```

```
      EXTERNAL         G05EYF
*     .. External Subroutines ..
      EXTERNAL         G05CBF, G05EDF, G05ZAF
*     .. Executable Statements ..
      CALL G05ZAF('O')
      WRITE (NOUT,*) 'G05EDF Example Program Results'
      WRITE (NOUT,*)
*
      CALL G05CBF(0)
*
      IFAIL = 0
*
      CALL G05EDF(N,P,R,NR,IFAIL)
*
      DO 20 I = 1, 5
         IX = G05EYF(R,NR)
         WRITE (NOUT,99999) IX
   20 CONTINUE
      STOP
*
99999 FORMAT (1X,I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05EDF Example Program Results

    54
    46
    48
    46
    56
```